

```

10 MODE 1
20 WINDOW £0,2,20,2,10:WINDOW £1,2,39,14,14
30 WINDOW £2,2,14,17,24:WINDOW £3,18,39,17,24
40 ORIGIN 350,240,350,620,380,240
50 INK 0,1:INK 1,6:INK 2,12:INK 3,22
60 PAPER £0,2:PAPER £1,3:PAPER £2,3:PAPER £3,2
70 PEN £0,1:PEN £1,0:PEN £2,0:PEN £3,1
80 CLS £0:CLS £1:CLS £2:CLS £3:CLG 1
90 display$="..... W I N D O W   £ 1....."
100 LOCATE £3, 2,2:PRINT £3,"Q U E U E   = ";
110 LOCATE £3,2,4:PRINT £3,"ENVELOPE   = ";
120 LOCATE £3,2,6:PRINT £3,"DURATION   = ";
130 color=1:y=0:shade=14:size=37:count=1:envelope=1
140 ENT 1,5,1,1,10,-1,1,5,1,1
150 ENT -2,10,2,3,17,-2,2,7,2,4
160 ENT -3,10,-2,2:ENT -4,10,-2,2
170 ENT -5,10,2,2,5,-4,1,20,1,4,5,-4,3
180 ENT -6,70,1,1,140,-1,1,70,1,1
190 EVERY 20,0 GOSUB 1000
200 EVERY 30,1 GOSUB 2000
210 EVERY 40,2 GOSUB 3000
220 EVERY 50,3 GOSUB 4000
230 LOCATE 5,1:PRINT"WINDOW £0"
240 LOCATE 6,3:PEN 1:PRINT"M E N U"
250 LOCATE 1,5:PEN 3:PRINT" 1 - Swap Windows "
260 LOCATE 1,6:PRINT" 2 - Change Border"
270 LOCATE 1,7:PRINT" 3 - Display ASCII "
280 LOCATE 2,9:PEN 3:PRINT"PRESS 1, 2 or 3"
290 a$=INKEY$:IF a$="1" THEN WINDOW SWAP 1,2:WINDOW SWAP 0,3
:GOTO 80
300 IF a$="2" THEN BORDER shade
310 shade= -shade*(shade<>25)+1
320 IF a$="3" THEN GOSUB 5000
330 IF flag=1 THEN flag=0:CLS £0:GOTO 230 ELSE 290
1000 REM This is the GRAPHIC WINDOW
1010 DI: ORIGIN 350,240:DRAW 640,y,color
1030 color= -color*(color<>3)+1
1040 y=y+5:IF y=1500 THEN CLG 1:y=0
1050 EI:RETURN
2000 REM This is WINDOW £1
2010 DI:MOVEIT$=RIGHT$(display$,size)+LEFT$(display$,37-size
)
2020 PRINT £1,MOVEIT$;:size=size-1
2030 IF size=0 THEN size=37
2040 EI:RETURN
3000 REM This is WINDOW £2
3010 PRINT £2,count;:count=(-count*(count<1E+09)*1.10332)+1
3020 RETURN
4000 REM This is the SOUND routine
4010 DI:free1=SQ(1) AND 7 :free2=SQ(2) AND 7:free3=SQ(4) AND
7
4020 IF free1 <>0 THEN READ channel,note,duration ELSE 4110
4030 LOCATE £3,14,2:PRINT £3,free1;free2;free3
4040 LOCATE £3,14,4:PRINT £3,envelope
4050 LOCATE £3,14,6:PRINT £3,duration;" "
4060 IF channel= -1 THEN RESTORE:GOTO 4110
4070 envelope= -envelope*(envelope<>6)+1
4080 SOUND channel,note,duration,,envelope
4090 DATA 1,200,200,4,300,200,7,700,80,1,178,50
4100 DATA 2,40,250,4,178,50,7,400,500,-1,-1,-1
4110 EI:RETURN
5000 REM Display ASCII characters
5010 flag=1:FOR character = 32 TO 255
5020 PEN £0,0: PRINT £0, CHR$(character);
5030 NEXT character
5040 PEN £0,3:PRINT:PRINT £0,"Press <ANY KEY>";
5050 WHILE k$="":k$=INKEY$:WEND
5060 k$="":RETURN

```